
;decodes.tw
;avance-version
;diffusion edited preparation period w/LED sequence
;homonuclear Hartman-Hahn transfer using MLEV17 sequence
;   for mixing
;using two power levels for excitation and spinlock
;phase sensitive using TPPI
;A. Bax & D.G. Davis, J. Magn. Reson. 65, 355-360 (1985)

;using gradient during SL for pure absorption phase
;see for ref. Titman et al.; JMR Ser. A 105, 167-183 (1993).

;TOCSY portion written by Dr. Katalin Kover

;DECODES expt. by Williamson R. T., Gerwick W. H. submitted to Org. Lett.

#include <Avance.incl>
#include <Grad.incl>

"p5=p6*.667"
"p7=p6*2"
;"d0=3u"
"d12=20u"
"d2=d6-p16-d16-p1-p1-d13"
"d13=4u"

1 ze
  d12 pl1:f1
2 d1
  50u UNBLKGRAD
  p1 ph1
   d13
   GRADIENT(cnst21)
   d16
  p1 ph1
   d2
  p1 ph1
   d13
   GRADIENT(cnst22)
   d16
  p1 ph1
   d26
  p1 ph1
  d0

  p16:f1:ngrad:c34 ph26
  d16
  p2 ph26
  d16 pl10:f1

4 (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph24 p7 ph25 p6 ph24)

  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph24 p7 ph25 p6 ph24)
  (p5 ph23)
  lo to 4 times l1

  p16:f1:ngrad:c34 ph26
  d16 pl1:f1
  p2 ph26
  d15                             ;d15+4u=d16=200u
  4u BLKGRAD

  go=2 ph31
  d11 wr #0 if #0 id0 ip1 zd
  lo to 2 times td1
exit

ph1=0 2 2 0 1 3 3 1

ph22=3 1 3 1 0 2 0 2
ph23=0 2 0 2 1 3 1 3
ph24=1 3 1 3 2 0 2 0
ph25=2 0 2 0 3 1 3 1
ph26=0 2 0 2 1 3 1 3
ph31=0 2 2 0 1 3 3 1

;grdprog  4sine
;cnst21=cnst22=variable
;cnst23=cnst24=20

;pl1 : f1 channel - power level for pulse (default)
;pl10: f1 channel - power level for TOCSY-spinlock
;p1 : f1 channel -  90 degree high power pulse
;p5 : f1 channel -  60 degree low power pulse
;p6 : f1 channel -  90 degree low power pulse
;p7 : f1 channel - 180 degree low power pulse
;p17: f1 channel - trim pulse                             [2.5 msec]
;d0 : incremented delay (2D)                         [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d12: delay for power switching                      [20 usec]
;L1: loop for MLEV cycle: (((p6*64) + p5) * l1) + (p17*2) = mixing time
;in0: 1/(2 * SW) = DW
;nd0: 2
;NS: 8 * n
;DS: 16
;td1: number of experiments
;MC2: TPPI
;d6 : encode  delay (2D)
;d26: delay for eddy current decay

;decodehet.tw

;avance-version
;2D H-1/X correlation via double inept transfer
;   using sensitivity improvement
;phase sensitive using Echo/Antiecho gradient selection
;with decoupling during acquisition
;using trim pulses in inept transfer
;A.G. Palmer III, J. Cavanagh, P.E. Wright & M. Rance, J. Magn.
;   Reson. 93, 151-170 (1991)
;L.E. Kay, P. Keifer & T. Saarinen, J. Am. Chem. Soc. 114,
;   10663-5 (1992)
;J. Schleucher, M. Schwendinger, M. Sattler, P. Schmidt, O. Schedletzky,
;   S.J. Glaser, O.W. Sorensen & C. Griesinger, J. Biomol. NMR 4,
;   301-306 (1994)

#include <Avance.incl>
#include <Grad.incl>

"p2=p1*2"
"p4=p3*2"
"d0=3u"
"d4=1s/(cnst2*4)"
"d11=30m"
"d13=3u"

"d2=d6-p16-d16-p1-p1-d13"

"d20=p16+d16+50u+p2+d0*2"
"d21=d13+p16+d16+4u"

"d31=p3-p1"

"l3=(td1/2)"
"ds=ns*2*cnst0"

1 ze
  d11 pl12:f2
2 d1 do:f2
  6m
3 d11
4 50u UNBLKGRAD
  p1 ph11
   d13
   GRADIENT(cnst21)
   d16
  p1 ph11
   d2
  p1 ph11
   d13
   GRADIENT(cnst22)
   d16
  p1 ph11

   d26
  p1 ph11
  d4 pl2:f2
  (d31 p2 ph1) (p4 ph1):f2
  d4
  p28 ph1
  d13
  (p1 ph2) (p3 ph3):f2
  d0
  p2 ph6
  d0
  50u ;UNBLKGRAD
  GRADIENT(cnst23)
  d16
  (p4 ph4):f2
  d20
  (p1 ph1) (p3 ph4):f2
  d24
  (d31 p2 ph1) (p4 ph1):f2
  d24
  (p1 ph2) (p3 ph5):f2
  d4
  (d31 p2 ph1) (p4 ph1):f2
  d4
  (p1 ph1)
  d21
  (p2 ph1)
  d13
  GRADIENT(cnst24)
  d16 pl12:f2
  4u BLKGRAD
  go=2 ph31 cpd2:f2
  d1 do:f2 wr #0 if #0 zd
  3m ip5
  3m ip5
  lo to 3 times 2
  d11 id0
  lo to 4 times l3
exit

ph1=0
ph2=1
ph3=0 2
ph4=0 0 2 2
ph5=1 1 3 3
ph6=0 0 2 2
ph11=0; 2 2 0 1 3 3 1
ph31=0 2 2 0

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse

;p3 : f2 channel -  90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p16: homospoil/gradient pulse
;p28: f1 channel - trim pulse
;d0 : incremented delay (2D)                  [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d4 : 1/(4J)XH
;d11: delay for disk I/O                             [30 msec]
;d13: short delay                                    [3 usec]
;d16: delay for homospoil/gradient recovery
;d20: = p16+d16+50u+p2+d0*2
;d21: = d13+p16+d16+4u
;d24: 1/(4J)XH for XH
;     1/(8J)XH for all multiplicities
;cnst0: ds = ns * 2 * cnst0
;cnst2: = J(XH)
;l3: loop for phase sensitive 2D using  E/A method : l3 = td1/2
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;NS: 1 * n
;DS: >= 16, but 2 * ns * m
;td1: number of experiments
;MC2: echo-antiecho
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

;d6 : encode  delay (2D)
;d26: delay for eddy current decay

;use gradient program (GRDPROG) : 4sineea_4

;use gradient ratio: cnst21 : cnst22 : cnst23 : cnst24
;       var :     var :     80 :    20   for C-13
;       var :     var :     80 :     8   for N-15

